Problem:
Score : 300 points

Problem Statement

There is a sequence of length N: A_1, A_2, ..., A_N. Initially, this sequence is a permutation of 1, 2, ..., N.

On this sequence, Snuke can perform the following operation:

Choose K consecutive elements in the sequence. Then, replace the value of each chosen element with the minimum value among the chosen elements.

Snuke would like to make all the elements in this sequence equal by repeating the operation above some number of times.
Find the minimum number of operations required.
It can be proved that, Under the constraints of this problem, this objective is always achievable.

Constraints

2 \leq K \leq N \leq 100000

A_1, A_2, ..., A_N is a permutation of 1, 2, ..., N.

Input

Input is given from Standard Input in the following format:

N K
A_1 A_2 ... A_N

Output

Print the minimum number of operations required.

Sample Input 1

4 3
2 3 1 4

Sample Output 1

2

One optimal strategy is as follows:

In the first operation, choose the first, second and third elements. The sequence A becomes 1, 1, 1, 4.

In the second operation, choose the second, third and fourth elements. The sequence A becomes 1, 1, 1, 1.

Sample Input 2

3 3
1 2 3

Sample Output 2

1

Sample Input 3

8 3
7 3 1 8 4 6 2 5

Sample Output 3

4

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

module ABC101

  ! <module>s to import
  use, intrinsic :: iso_fortran_env

  ! require all variables to be explicitly declared
  implicit none

  ! accessibility of <subroutine>s and <function>s in this <module>
  public  :: task_C
  private :: calculate_min_num_operations

  ! contained <subroutine>s and <function>s are below
  contains

  subroutine task_C

    ! variables for this <subroutine>
    integer(INT32) :: len_sequence
    integer(INT32) :: num_elements_replacing

    ! STEP.01
    ! read out ...
    ! 1. the length of the given sequence
    ! 2. the number of elements to replace at once
    read(unit=INPUT_UNIT, fmt=*) len_sequence, num_elements_replacing

    ! STEP.02
    ! calculate & output minumum value of the number of operations that can be realized in a given sequence
    write(unit=OUTPUT_UNIT, fmt='(I0)', advance='yes') calculate_min_num_operations(len_sequence, num_elements_replacing)

    ! STEP.END
    return

  end subroutine task_C

  pure function calculate_min_num_operations (len_sequence, num_elements_replacing) result(num_operations)

    ! arguments for this <function>
    integer(INT32), intent(in) :: len_sequence
    integer(INT32), intent(in) :: num_elements_replacing

    ! return value of this <function>
    integer(INT32) :: num_operations

    num_operations &!
      = ceiling( &!
        real(len_sequence           - 1_INT32, kind=REAL64) / &!
        real(num_elements_replacing - 1_INT32, kind=REAL64),  &!
        kind=INT32                                            &!
      )

      return

  end function calculate_min_num_operations

end module ABC101


program main

  ! <module>s to import
  use, non_intrinsic :: ABC101

  ! require all variables to be explicitly declared
  implicit none

  call task_C

end program main